How to find the centroid of a region between two functions

In mathematics and physics, the *centroid* or *geometric center* of a plane figure is the arithmetic mean position of all the points within the shape.

The coordinates of the centroid of standard 2-dimensional geometric objects such as triangles, parallelograms, quarter/half circles, ellipses, etc. are well-known [1], [2], [3].



If the object is bounded by the graph of two continuous functions f and g, the calculation of the centroid coordinates is more elaborate:

The centroid **C (x_c | y_c)** of a region bounded by the graphs of the continuous functions f and g such that f (x) ≥ g (x) or g (x) ≥ f (x) on the interval [a, b] with a ≤ x ≤ b, is given by

$$x_c = \frac{1}{A}\int_a^b x\,[f(x) - g(x)]\,dx$$

$$y_c = \frac{1}{2\,A}\int_a^b [\,[f(x)]^2 - [g(x)]^2]\,dx$$

where A is the area of the region:

$$A = \int_a^b [f(x) - g(x)]\,dx$$

The above condition for f and g claims that f and g must not intersect within the interval [a, b].

The GC3 file **Centroid.gc3** shows a plot of the functions f and g, the region bounded by the graphs and the borders a and b, and the centroid. Furthermore, it alerts you if f and g intersect within the interval [a, b].



Programming Details:

- The three integrals are computed numerically with the extended Simpson Rule (for more details refer to **How to calculate integrals with GC3 / 3. Numerical Integration**).
- The coloring for the region between f and g within the interval [a, b] is done in an 'inverted' way, i.e. the region itself is 'empty' whereas the outer region in the interval is white. You will find more details and an alternative in **How to calculate integrals with GC3 / 5. Area between two functions**.
- The centroid C (x_c | y_c) can be displayed as crosshairs and/or dotted lines:

```
y=  yc   "dotted line
x=  xc   "dotted line
sc=0.4   "size of crosshairs marking the centroid        0.4
xcross(x)=yc; x>=xc-sc and x<=xc+sc
ycross(x)=xc; y>=yc-sc and y<=yc+sc
y=  xcross(x)   "crosshairs
x=  ycross(y)   "crosshairs
```

- The check for an intersection of f and g within [a, b] is more elaborate and done in the following way:
  - For the difference function h (x) = f (x) – g(x) the minimum y_{min} and maximum y_{max} are computed.

    This is done by dividing the interval [a, b] into N parts (e.g. N=500) and letting a recursion over all parts find the minimum and maximum y value:

```
"---------- Check for intersection(s) of f and g within [a, b]
N=500  "number of data points in [a,b]                    500
delta_x=(b-a)/N                                           0.02
h(i)=f(a+i*delta_x)-g(a+i*delta_x) "discrete difference function for f(x)-g(x)
"find maximum M and minimum m of h within [a, b]
M(j)=max(h(j), M(j-1)) ;j>0
m(j)=min(h(j), m(j-1))   ;j>0
M(j)=h(j)                ;j=0
m(j)=h(j)                ;j=0
ymax= M(N)                                                5.92590816
ymin=m(N)                                                 0
```

  - By comparing the algebraic signs of y_{min} and y_{max}, a function *intsec_flag* delivers either 1 if there is an intersection or 0 if there is no intersection, following these conditions:

$$intsec\_flag = \begin{cases} 1, & \text{sign}(y_{min}) \neq \text{sign}(y_{max}) \text{ and } \text{sign}(y_{min}), \text{sign}(y_{max}) \neq 0 \\ 0, & \text{sign}(y_{min}) = \text{sign}(y_{max}) \\ 0, & \text{sign}(y_{min}) \text{ or } \text{sign}(y_{max}) = 0 \end{cases}$$

    In other words: f and g do not intersect if both of the algebraic signs are equal or one of them is 0 which is implemented in the following way:

```
signum(x)=sign(x) ;abs(x)>=10^(-15)  "modified sign-function
signum(x)=0       ;abs(x)<=10^(-15)
signum(ymax)                                  1
signum(ymin)                                  0
intsec_flag(dummy)=0 ;signum(ymax)*signum(ymin)>=0
intsec_flag(dummy)=1 ;signum(ymax)*signum(ymin)<0
intsec_flag(0)                                0
```

Note: Instead of using the built-in function 'sign', a modified function 'signum' is used. This is necessary since it may happen that the calculation of y_{min} or y_{max} delivers a 0 but the sign function delivers a 1 or -1 due to calculation accuracy.

- The variable intsec_flag is then used to colorize the plot area red if there is an intersection (intsec_flag = 1), alerting you that the displayed point C is not valid anymore:

```
check(dummy)=x      ;intsec_flag(0)=1
y>   check(0)  "red xy plane if intersection within [a, b]
```

In this example, the value b for the right border is set too high so that an intersection takes place within [a, b]:



As an alternative or addition to this you may suppress displaying C and the dotted lines as shown here for C:

```
xcross(x)=yc; x>=xc-sc and x<=xc+sc
xcross1(x)=xcross(x); intsec_flag(0)=0
ycross(x)=xc; y>=yc-sc and y<=yc+sc
ycross1(x)=ycross(x); intsec_flag(0)=0
y=   xcross1(x)  "crosshairs
x=   ycross1(y)  "crosshairs
```

Note: The additional lines 'xcross1(x)=…' and 'ycross1(x)=…' are necessary since GC3 can 'only' handle two expressions combined with AND but three are needed.

Further Examples

- Find the centroid of the region between f and g within [0, 2π] with

  f (x) = [sin (x)] ² +2.5

  g (x) = - [sin (x)] ² +2.5

  Note: f and g do not intersect at point (π | 2.5); they only touch each other at this point.



- Find the centroid of the region between f and g within [0, 2π] with

  f (x) = sin (x) +2.5

  g (x) = - sin (x) +2.5

  Note: At point (π | 2.5) f and g intersect and a warning is displayed.



- Find the centroid of a quarter circle with a radius of 1:

  $f(x) = \sqrt{1 - x^2}$

  g (x) = 0

  Interval: [0, 1]

  Exact values:

  $$x_c = y_c = \frac{4}{3\pi} \qquad A = \frac{\pi}{4}$$



Comparing the exact values with the calculated values shows an absolute error of approx. 10^{-6}. This is due to the fact that f has an infinite slope at point (1 | 0). (compare with **How to calculate integrals with GC3 / 3. Numerical Integration**).

Quick References:

[1] https://en.wikipedia.org/wiki/Centroid

[2] https://en.wikipedia.org/wiki/List_of_centroids

[3] http://www.bgu.ac.il/~yakhot/mf1/Centroids.pdf